



# A Spice-Based Code for ARL's 4.5-MJ Electromagnetic Launcher Pulsed Power Supply System

by Miguel Del Güercio

ARL-TR-2592

September 2001

Approved for public release; distribution is unlimited.

20020129 109

The findings in this report are not to be construed as an official Department of the Army position unless so designated by other authorized documents.

Citation of manufacturer's or trade names does not constitute an official endorsement or approval of the use thereof.

Destroy this report when it is no longer needed. Do not return it to the originator.

# Army Research Laboratory

Aberdeen Proving Ground, MD 21005-5066

---

ARL-TR-2592

September 2001

---

## A Spice-Based Code for ARL's 4.5-MJ Electromagnetic Launcher Pulsed Power Supply System

Miguel Del Güercio

Weapons and Materials Research Directorate, ARL

---

Approved for public release; distribution is unlimited.

---

---

## Abstract

---

A Spice-based code (Ispice/4Rx) is used to simulate the U.S. Army Research Laboratory's (ARL's) 4.5-MJ pulsed power supply (PPS) discharge into an electromagnetic (EM) railgun. The code determines the in-bore projectile position, voltage, and current at the breech, and exit current and velocity at the muzzle exit. The code also includes a diffusion model of the railgun rails. The code's primary inputs are the charging voltage to the PPS capacitor banks and the parameters for the different inductive and resistive components of the system which model its hardware. The railgun parameters were chosen to reflect current hardware and program requirements.

---

## Acknowledgments

---

The author was supported by consultations with Mr. Trevor Watt from the Institute for Advanced Technology (IAT), The University of Texas at Austin.

INTENTIONALLY LEFT BLANK.

---

## Contents

---

Acknowledgments	iii
List of Figures	vii
List of Tables	ix
1. Introduction	1
2. Code Structure	3
3. Code Validation	10
4. Code Output	10
5. Conclusions	16
6. References	17
Appendix A. Netlist of 4.5-MJ Spice Code With Railgun Load	19
Appendix B. Schematics of 4.5-MJ Spice Code With Railgun Load	31
Distribution List	41
Report Documentation Page	45

INTENTIONALLY LEFT BLANK.



---

## List of Figures

---

Figure 1. 4.5-MJ racks.....	2
Figure 2. Schematic example. ....	3
Figure 3. Railgun functions.....	4
Figure 4. Netlist statements for DEFINE, PRINT, and TRAN (transient analysis).....	5
Figure 5. Delay generator subcircuit netlist.....	6
Figure 6. Voltage-controlled switch S1 with netlist model.....	7
Figure 7. Call statements for modules and subcircuits.....	8
Figure 8. Railgun call statement. ....	8
Figure 9. Railgun subcircuit.....	9
Figure 10. Peak current vs. module inductance.....	11
Figure 11. Module output at 11 kV with 60- $\mu$ H inductors into a 1-m $\Omega$ load. ....	11
Figure 12. Module current at 9 kV with 60- $\mu$ H inductors. ....	12
Figure 13. Position, velocity, and gun current at 9 kV with 60- $\mu$ H inductors. ....	12
Figure 14. Module output at 11 kV with 60- $\mu$ H inductors. ....	13
Figure 15. Position, velocity, and gun current at 11 kV with 60- $\mu$ H inductors. ....	13
Figure 16. Module output for a short circuit at the breech (MASS=9999 kg, V [12] and V [20] = 0). ....	14
Figure 17. Gun current for a short circuit at the breech (MASS=9999 kg, V [12] and V [20] = 0).....	14
Figure 18. Module output at 9 kV with 60- $\mu$ H inductors into a 1-m $\Omega$ load.....	15
Figure B-1. Bank 1/module 1; X <sub>11 11 0 14</sub> calls subckt module 1 <sub>243</sub> . ....	31
Figure B-2. Bank 1/module 2; X <sub>12 12 0 24</sub> calls subckt module 2 <sub>243</sub> . ....	31
Figure B-3. Bank 2/module 3; X <sub>13 13 0 34</sub> calls subckt module 3 <sub>243</sub> . ....	32
Figure B-4. Bank 2/module 4; X <sub>14 100 0 44</sub> calls subckt module 4 <sub>243</sub> . ....	32
Figure B-5. Bank 3/module 5; X <sub>15 15 0 54</sub> calls subckt module 5 <sub>243</sub> . ....	33
Figure B-6. Bank 3/module 6; X <sub>16 16 0 64</sub> calls subckt module 6 <sub>243</sub> . ....	33
Figure B-7. Bank 4/module 7; X <sub>17 17 0 74</sub> calls subckt module 7 <sub>243</sub> . ....	34

Figure B-8. Bank 4/module 8; $X_{18\ 18\ 0\ 84}$ calls subckt module 8 <sub>243</sub> . .....	34
Figure B-9. Bank 5/module 9; $X_{1\ 1\ 0\ 94}$ calls subckt module 9 <sub>243</sub> . .....	35
Figure B-10. Bank 5/module 10; $X_{2\ 20\ 0\ 104}$ calls subckt module 10 <sub>243</sub> . .....	35
Figure B-11. Bank 6/module 11; $X_{3\ 130\ 0\ 114}$ calls subckt module 11 <sub>243</sub> . .....	36
Figure B-12. Bank 6/module 12; $X_{4\ 40\ 0\ 124}$ calls subckt module 12 <sub>243</sub> . .....	36
Figure B-13. Bank 7/module 13; $X_{5\ 150\ 0\ 134}$ calls subckt module 13 <sub>243</sub> . .....	37
Figure B-14. Bank 7/module 14; $X_{6\ 60\ 0\ 144}$ calls subckt module 14 <sub>243</sub> . .....	37
Figure B-15. Bank 8/module 15; $X_{7\ 70\ 0\ 154}$ calls subckt module 15 <sub>243</sub> . .....	38
Figure B-16. Bank 8/module 16; $X_{8\ 80\ 0\ 164}$ calls subckt module 16 <sub>243</sub> . .....	38
Figure B-17. Bank 9/module 17; $X_{9\ 90\ 0\ 174}$ calls subckt module 17 <sub>243</sub> . .....	39
Figure B-18. Bank 9/module 18; $X_{10\ 100\ 0\ 84}$ calls subckt module 18 <sub>243</sub> . .....	39
Figure B-19. Railgun outer model B-functions. ....	40
Figure B-20. Railgun inner model B-functions. ....	40
Figure B-21. Railgun diffusion ladder. ....	40

---

## List of Tables

---

Table 1. Summary of relevant modeling results. ....	16
---	----

INTENTIONALLY LEFT BLANK.

---

## 1. Introduction

---

A 4.5-MJ pulsed power system (PPS) previously operated by GDLS (General Dynamics Land Systems) on electrothermal-chemical (ETC) programs was acquired by the U.S. Army Research Laboratory - Weapons and Materials Research Directorate (ARL-WMRD) and is currently being modified to operate with an EM railgun. Simulations of this pulser's operations were generated with the Spiced-based code Intusoft ICAP/4Rx Software (Intusoft, San Pedro CA). The software uses a variable time step control algorithm which reduces the usually long run times involved in these types of simulations.

The pulser's hardware consists of eighteen 250-KJ independently triggered modules [1]. Pairs of these modules are symmetrically placed in nine racks (3 ft 7 in wide  $\times$  13 ft long and 4 ft high). Each of the eighteen 250-KJ modules contains five Aerovox 11 kV 50-KJ capacitors and two 116- $\mu$ H inductors. The two inductors can be connected as a single 116- $\mu$ H inductor or in parallel as a 60- $\mu$ H inductor. There is a 4-m $\Omega$  resistance per single 116- $\mu$ H inductor, and a 2-m $\Omega$  resistance per single 60- $\mu$ H inductor, approximately. A total of 15 diodes arranged as five in parallel, with three in series in each pack (DSA 908 Avalanche Diode), form a crowbar diode set which protects the capacitor's from damaging voltage reversals. Sets of four fuses in parallel, in series with the output of each of the capacitor terminals, enable the handling of large currents while isolating the rest of the circuitry in case of a capacitor short circuit at high voltages. Each module discharge is initiated by a high voltage trigger generator (PI-TG-75S) and controlled by a spark gap (PI-ST-300) operating as an output switch [2].

The system operation is remotely monitored from a consol control which incorporates state-of-the-art fiber optic linked subsystems. The consol control contains six separate chassis: (1) the control chassis with the on/off system key switch and enable press buttons, (2) the charge monitor panel displaying the different real-time readings, (3) the charge monitor chassis, (4) the output monitor chassis, (5) an 18-channel delay generator which allows the modules to be fired at preselected times, and the (6) interlock chassis which defeats any attempt at discharging the pulser if any interlock is not closed.

Initially, the modules also included a 6-m $\Omega$  resistor in series with the crowbar diode stack which becomes vulnerable when switching at large voltages. As data for this configuration was available, the code initial module input parameters included this 6-m $\Omega$  resistor (RCB) with the purpose of validating the code's output. These code calculations show RCB as 6.65 m $\Omega$ , as it includes a series 0.65 m $\Omega$  bus resistance. However, this resistor is not installed in the final and present pulser's configuration. Instead, five 15 m $\Omega$  resistors in series with each

of the five diode stacks and in parallel to each other provide a total of 3 m $\Omega$  crowbar resistance per module. This configuration was not documented and replaces the previous 6 m $\Omega$  resistor.

The time for charging this system to full voltage (11 kV) is about 60 s, with a maximum current of 15 Amps. Initially, the charging starts in a constant current mode, and its value is dialed in a local controller for the HVDC (high voltage DC) power supply. Once the set point voltage is reached, the HVDC power supply switches to a constant voltage mode and regulates while holding the set voltage. The delay generator triggers within a 100-ms window, and the system fires within 250 ms. However, if the system is not fired within 90 s, an excess time counter, started when the key switch is turned to run, will connect resistors across the capacitors to deplete the energy stored in the capacitors. The pulser's discharge is conducted through 18 single-coaxial, 35-ft-long cables (160  $\mu\Omega$ /ft), which connect each module's output in parallel to the breech of the railgun. Figure 1 shows the 4.5-MJ racks, while preparatory work was in process to complete its final assembly and configuration.




Figure 1. 4.5-MJ racks.

Since Spice is not specifically designed to handle mechanical terms, one option would be to rewrite all the kinematics equations involved in the railgun modeling in circuit form. The other option is to include nonlinear source elements where the mechanical terms can be defined in their own domain. For

this reason, the code makes extensive use of nonlinear sources, or "B" functions [3]. The B functions allow the modeling of nonlinear elements by using their linear models through a change of variable with the nonlinear-dependent model source; in other words, the B functions are a function within a function. The notation for these functions begins with B, and these notations can be either current or voltage generating. As there is no distinction between a current controlled or a voltage controlled nonlinear dependent source, the source is defined as current controlled if "I" is specified, and similarly voltage controlled if "V" is given.

Due to the difficulty of generating schematics representing the B functions that describe the railgun, the code was generated instead by first writing a "netlist." The code includes a diffusion model of the railgun [3] to more accurately simulate the resistive losses in the rails. Figures 2(a) and 2(b) show an example of the schematics for the first bank's two modules. Figures 3(a-c) show the railgun schematics and B functions. The complete code is shown in Appendix A, and a list of all the schematics in Appendix B.





Figure 2. Schematic example.


## 2. Code Structure

The following analysis of the code commands and nomenclature is presented by analyzing segments of the total code for sake of clarity (Figure 4). Parameters of interest are initially defined; they are voltage and current at the breech ( $VBREECH=V(22)$ ,  $IBREECH=I(V2)$ , muzzle voltage ( $VMUZZLE=V(35)$ ), velocity ( $VELOCITY=V(31)$ ), projectile position ( $POSITION=V(32)$ ), and force ( $FORCE=V(33)$ ). Auxiliary current-controlled voltage sources are also defined (i.e., VM1...VM18) and are positioned across the modules inductors impedance


(RC1... RC18) to measure the modules output currents without affecting the circuit's operation. The total pulser's output current is measured similarly by "V2," placed between nodes 23 and 30 (see Figure 3[a], railgun/B functions).



a. Railgun outer model B-functions.



b. Railgun inner model B-functions.



c. Diffusion ladder for rails.

Figure 3. Railgun functions.

The statement ".PRINT TRAN I(VM1),...,I(VM18), I(V2)" saves the value of the current through the voltage source during the transient analysis. The command "save allcur" and save "allpow" allows the plots of all the currents or power through all the elements involved in the analysis. This option slows the processing time substantially and is only used during the final simulations. The DEFINE statements characterize the fixed parameters, VCHARGE is the initial voltage of the banks, LPRIME is the inductance gradient of the launcher, MASS is the projectile mass, and XSTART is the initial axial position of the projectile. The rail separation times the rail height is defined as the cross sectional area of the bore; however, this area is estimated as less than the actual value to compensate for armature inefficiencies during launch.



---

#### 4.5 MJ PULSER CODE WITH RAILGUN LOAD NETLIST

```
.control
alias VBREECH=V(22)
alias IBREECH=I(V2)
alias VMUZZLE=V(35)
alias VELOCITY=V(31)
alias POSITION=V(32)
alias FORCE=V(33)
alias IPKMOD1=I(VM1)
alias IPKMOD2=I(VM2)
alias IPKMOD3=I(VM3)
alias IPKMOD4=I(VM4)
alias IPKMOD5=I(VM5)
alias IPKMOD6=I(VM6)
alias IPKMOD7=I(VM7)
alias IPKMOD8=I(VM8)
alias IPKMOD9=I(VM9)
alias IPKMOD10=I(VM10)
alias IPKMOD11=I(VM11)
alias IPKMOD12=I(VM12)
alias IPKMOD13=I(VM13)
alias IPKMOD14=I(VM14)
alias IPKMOD15=I(VM15)
alias IPKMOD16=I(VM16)
alias IPKMOD17=I(VM17)
alias IPKMOD18=I(VM18)
save (22)I(V2)I(VM1)I(VM2)I(VM3)I(VM4)I(VM5)I(VM6)I(VM7)I(VM8)I(VM9)V(22)I(V2)I(VM10)I(VM11)I(VM12)I(VM13)I(VM14)
+I(VM15)I(VM16)I(VM17)I(VM18)
**SAVE ALL ALLCUR
**SAVE ALL ALLCUR ALLPOW
.endc
*DEFINE VCHARGE=9K
*DEFINE LPRIME=6.20E-7
*DEFINE MASS=0.190
*DEFINE XSTART=0.3
*DEFINE AREA=2.0E-3
*DEFINE RARM=1.8E-5
*DEFINE VARM=0
*DEFINE INTVEL=0
.OPTIONS TEMP=27 ISCALE=1000000 VSCALE=3000
.TRAN 10U 10M 0 5U UIC
.PRINT TRAN I(V2) VBREECH IBREECH VMUZZLE VELOCITY POSITION FORCE
*.PRINT TRAN V(22)V(35)V(31) V(32) V(33)I(V2)I(VM1)I(VM17)
.PRINT TRAN V(31)I(V2)I(VM1)I(VM2)I(VM3)I(VM4)I(VM5)I(VM6)I(VM7)I(VM8)I(VM9)I(VM10)
+I(VM11)I(VM12)I(VM13)I(VM14)I(VM15)I(VM16)I(VM17)I(VM18)
* The delay generator determines the banks firing times by closing the switch S1 at each bank
* and putting a voltage of 9kV at nodes 14,24,34,...184 where the V1 voltage source is located
* for each case.
X15 14 24 34 44 54 64 74 84 94 104 114 124 134 144 154 164 174 184 DLYGEN {T1=0.0
+ T2=0.0 T3=0.0U T4=0.0U T5=0.0U T6=0.0U T7=0.0U T8=0.0U T9=0.0U T10=0.0U T11=0.0U
+T12=0.0U T13=0.0U T14=0.0U T15=0.0U T16=0.0U T17=0.0U T18=0.0U}
```

---

Figure 4. Netlist statements for DEFINE, PRINT, and TRAN (transient analysis).

The .TRAN statement specifies the transient analysis values for the parameters TSTEP, TSTOP, TSTART, and TMAX, which control the output data step, the total analysis time, start time for analysis, and the maximum internal time step, respectively. The use initial conditions (UIC) statement at the end specifies that the ICAP/4Rx code will ignore an initial transient solution which takes place before the transient analysis starts. However, the code will then use the IC=values and the IC statements only.

Call statements for the subcircuits begin with X, and the subcircuit name and subnames are specified after the node list "Xname N1 N2 N3...subname." The order of nodes in the calling statement X agrees with the nodes nomenclature and order of the subcircuit statement (.SUBCKT). The list of components in each of the 18 module models after each SUBCK MODULE# statement is identical, except for the value of the capacitance CB.

X15 is the call for the subcircuit DLYGEN (Figure 5). The .SUBCKT DLYGEN models the function of the trigger generator located in the consol control, which triggers each spark gap switch (ST-300) corresponding to each of the 18 modules. This operation is modeled through a pulse or transient signal generator.

---

```

SUBCKT DLYGEN 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 {T1=0 T2=0 T3=0 T4=0
+ T5=0 T6=0 T7=0 T8=0 T9=0 T10=0 T11=0 T12=0 T13=0 T14=0 T15=0 T16=0 T17=0 T18=0}
V1 1 0 PULSE 0 300 {T1} 1U 1U 10U
V2 2 0 PULSE 0 300 {T2} 1U 1U 10U
V3 3 0 PULSE 0 300 {T3} 1U 1U 10U
V4 4 0 PULSE 0 300 {T4} 1U 1U 10U
V5 5 0 PULSE 0 300 {T5} 1U 1U 10U
V6 6 0 PULSE 0 300 {T6} 1U 1U 10U
V7 7 0 PULSE 0 300 {T7} 1U 1U 10U
V8 8 0 PULSE 0 300 {T8} 1U 1U 10U
V9 9 0 PULSE 0 300 {T9} 1U 1U 10U
V10 10 0 PULSE 0 300 {T10} 1U 1U 10U
V11 11 0 PULSE 0 300 {T11} 1U 1U 10U
V12 12 0 PULSE 0 300 {T12} 1U 1U 10U
V13 13 0 PULSE 0 300 {T13} 1U 1U 10U
V14 14 0 PULSE 0 300 {T14} 1U 1U 10U
V15 15 0 PULSE 0 300 {T15} 1U 1U 10U
V16 16 0 PULSE 0 300 {T16} 1U 1U 10U
V17 17 0 PULSE 0 300 {T17} 1U 1U 10U
V18 18 0 PULSE 0 300 {T18} 1U 1U 10U
.ENDS

```

---

Figure 5. Delay generator subcircuit netlist.

Each pulse is defined by the corresponding module voltage source V, the initial pulse value (0V) and the final value (300V), the delay time {T} (which can be different from zero), the rise time (1  $\mu$ s), the fall time (1  $\mu$ s) and the pulse width (10  $\mu$ s). Subsequently, the energized voltage sources V1....V18 enable the spark gap switches modeled by switch S1 model SMOD1 SW. When the ST-300 is triggered, the energy stored in the capacitor CB will flow to the circuit. For Module1, this process starts as X1 calls the subcircuit MODULE1; the voltage-controlled switch S1, which models the spark gap, will close as it receives its ON voltage from source V1 energized at time T = 0 (Figure 6). Switch S1 nomenclature is characterized in the netlist by four digits, where the first two digits represent the nodes from which the switch is connected to the circuit (nodes 7 and 8), followed by the nodes that connect the switch to the voltage source controlled by the delay generator (nodes 3 and 0). This voltage-controlled switch .MODEL SMOD1 SW (with hysteresis) is in the ON state at voltage VT+VH with a resistance RON, and is in the OFF state at voltage VT-VH with a resistance ROFF.

---

```
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
```

---

Figure 6. Voltage-controlled switch S1 with netlist model.

Each module crowbar diode assembly is modeled through a Spice switching diode model DSA908 D, with parameters adjusted to those of the avalanche DSA908 diode installed in the actual crowbar diode module assembly. The resistance of the model (0.264 m $\Omega$ ) is the equivalent of the 15 crowbar diodes, as each diode has a nondynamic resistance of 0.44 m $\Omega$ .

The call statement X11 of the first module, besides specifying the values of the module's output coaxial cable inductance and impedance values (i.e., L1C and R1C) calls its SUBCKTMODULE1. This in turn provides the capacitance value, bus parameters, crowbar diode modeling, spark gap modeling, and the module's inductor parameters (LIND and RIND) (Figure 7).

Immediately after the last call statement X10 in module 18, the railgun subroutine X20 30 31 32 33 35 RAILGUN (Figure 8) calls .SUBCKT RAILGUN 1 2 12 13 14 15. The order of nodes in the calling statement X20 is the same as the node's nomenclature and order of the nodes in the subcircuit statement.

---

```

X11 11 0 14 MODULE1 {V1=VCHARGE}
L1C 11 201 1.2U
VM1 11 201 0
R1C 201 22 3.18M

.....

.SUBCKT MODULE1 2 4 3 {V1= VCHARGE}
CB 5 4 4242.4U IC={V1}
*RS is 3M for the Capacitor resistance + 0.65M for the 4 Fuse parallel pack.
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7 M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS

```

---

Figure 7. Call statements for modules and subcircuits.

---

```

X10 100 0 184 MODULE18 {V18=VCHARGE}
L18C 100 213 1.2U
VM18 100 213 0
R18C 213 22 3.18M
*Railgun nodes/ LBC and RBC are the connection's inductance and impedance to the railgun.
LBC 22 239 0.25U
RBC 239 23 .5M
V2 23 30 0
X20 30 0 31 32 33 35 RAILGUN

```

---

Figure 8. Railgun call statement.

The BB subcircuit railgun expression places this B function between nodes 10 and 2 (see Figure 3[a]). All the calculations in the subcircuit are based on the current input through node 10, with BB as the connection between the subcircuit and the subcircuit's modules. These calculations include all the power loss terms of the railgun model and railgun current diffusion calculated in the inner modeling (see Figures 3[b] and 3[c]). After this iteration, BB connects back to the outer model (Figure 3[a]) before it gets grounded.

VB reads the current going into BB (Figure 9), and all the other railgun subcircuit elements also read their input current from VB (see Figures 3[a] and 3[c]). The inductor circuit containing LB approximates the derivative of the current I, and the RC circuits perform integrations.

---

```

.SUBCKT RAILGUN 1 2 12 13 14 15
BB 10 2 V= LPRIME*I(VB)*V(12)+(V(20)+2.13e-6*I(VB)*(70+1/SQRT(V(30)+1.E-5)))*(V(13)+ XSTART) + V(15)
*****
*BB 10 2 V= (2.13e-6*I(VB)*(70+1/SQRT(V(30)+1.E-5)))*(V(13)+ XSTART) + V(15)
*****

VB 1 10 0
* Voltage source to monitor the input current
FB 0 11 VB 1
LB 11 0 1
* Convert input current to a current and apply to a 1 Henry inductor.The
* voltage v(11) is I-dot.
BA 14 0 V= LPRIME*I(VB)^2/2
* Voltage v(14) is the driving force.
BV 0 12 I= (LPRIME*I(VB)^2/2 - 1.44*V(12)^2* AREA)/ MASS
CV 12 0 1 IC= INTVEL
RV 12 0 1E6
* A current source BV, proportional to acceleration, charges CV. The
* voltage V(12) is the velocity.
BX 0 13 I=V(12)
CX 13 0 1
RX 13 0 1E6
* A current source BX, proportional to velocity, charges CX.
* voltage v(13) is the position.
BM 15 0 V= RARM*I(VB) + VARM
* source BM produces the muzzle voltage on V(15).
FR 0 20 VB 1
LR 20 21 LPRIME
R1 21 0 1.75M
L1 21 22 9.72N
R2 22 0 0.88M
L2 22 23 9.72N
R3 23 0 0.58M
L3 23 24 16.5N
R4 24 0 0.44M
L4 24 25 14.6N
R5 25 0 0.29M
L5 25 26 23.3N
R6 26 0 0.22M
L6 26 27 15.6N
R7 27 0 0.18M
L7 27 28 8.7N
R8 28 0 0.29M
* current source FR drives the input current through a diffusion
*model of the rails. The voltage V(20) is the inductive and resistive voltage per meter.
IT 0 30 1
CT 30 0 1 IC=0
*Current source IT charges CT with 1 Amp. Voltage V(30) is equal to time.
.ENDS
.END

```

---

Figure 9. Railgun subcircuit.

The RC circuits first integrate the acceleration, which is defined as the following.

$$I = (\text{Electromotive Force (emf)} - \text{Drag Force}) / \text{MASS} \quad (1)$$

$$\text{emf} = 1/2 \text{ LPRIME} * I(VB)^2 / 2 \quad (2)$$

$$\text{Drag Force} = 1.44 * V(12)^2 * \text{Area} \quad (3)$$

This acceleration is then integrated to obtain the velocity represented as the resultant voltage across the capacitor CV. The voltage is integrated again to obtain the position represented as the voltage across the capacitor CX.

Finally, a current source FR which gets its input from VB, drives the current through the inductive and resistive diffusion ladder network where the inductive and resistive gradient impedance loss over time is calculated. The distance that the armature has moved along the rails is then multiplied by voltage to calculate the inductive or resistive impedance.

The time variable is simulated by the voltage across the capacitor CT with a constant current of 1 Amp. The resultant voltage is then evaluated as the product of the current of 1 Amp times the "time" variable, which renders "time."

---

### 3. Code Validation

---

A preliminary evaluation of this code was done for a charge of 11 kV with 60-μH inductors into a 1-mΩ load, which replaced the railgun (Figure 10). A peak output current of 1.6 MA for the total output of the 18 modules at 89 kA per module was in good agreement with 1.62 MA and 90 kA reported by GDLS tests [1]. Also shown in Figure 10 are peak current values expected at 9 kV under short circuit conditions with different values of module inductance. A code simulation at 11 kV with 60-μH inductors into a 1-mΩ load (Figure 11) closely matched the 90 kA per module or 1.62-MA total output value obtained on GDLS tests.

---

### 4. Code Output

---

Simulations for the railgun load cases (Figures 12–15) were performed at two different charging voltages (9 kV and 11 kV). Also, a short circuit at the breech of the railgun is considered (Figures 16 and 17), where the terms V(12) and V(20) of the railgun subcircuit, together with an increased mass, were used to model a stationary armature. Assumed values include the railgun length (2.87 m), the inductance gradient (0.6 μH/m), the armature resistance (18 μΩ), the initial




Figure 10. Peak current vs. module inductance.




Figure 11. Module output at 11 kV with 60- $\mu$ H inductors into a 1-m $\Omega$  load.




Figure 12. Module current at 9 kV with 60- $\mu$ H inductors.




Figure 13. Position, velocity, and gun current at 9 kV with 60- $\mu$ H inductors.






Figure 14. Module output at 11 kV with 60- $\mu$ H inductors.




Figure 15. Position, velocity, and gun current at 11 kV with 60- $\mu$ H inductors.




Figure 16. Module output for a short circuit at the breech (MASS=9999 kg, V [12] and V [20] = 0).




Figure 17. Gun current for a short circuit at the breech (MASS=9999 kg, V [12] and V [20] = 0).

armature position (0.3 m), and a 190-g launch mass. Figure 12 shows a typical module current output at 9 kV. The current exponentially decays with a time constant of 1.9 ms due to the resistance of the added 6-m $\Omega$  resistor in series with the crowbar diode sets.

Plot no. 1 in Figure 13 shows the armature at the muzzle at 2.31 ms having traveled 2.87 m. The projectile's exit velocity of 2260 m/s at 2.31 ms is shown on plot no. 2, indicating a current of 471 kA (plot no. 3) when the armature exits the launcher.

The module current at 11 kV with the railgun load is shown in Figure 14. Figure 14 shows module behavior similar to Figure 12, with a current output of 73.8 kA at 11 kV. Plot no. 3 in Figure 15 shows the armature position at the muzzle at 1.92 ms (2.87 m). The projectile's exit velocity of 2930 m/s at 1.92 ms is shown on plot no. 1, and plot no. 2 shows a current of 656 kA at the muzzle.

To simulate a short circuit at the breech, the projectile mass was increased to a very large number (i.e., 9999 kg). Also, the term V(12), which represents the velocity, and V(20), which calculates the power loss in the diffusion ladder of the railgun, were made equal to zero. Figure 16 shows a peak output of 78.7 kA for a module charged to 9 kV with a short circuit at the breech. Figure 17 shows the gun current under the same initial conditions.

The simulations at 9 kV (Figure 18) illustrate that higher module currents occur when discharging into a short at the breech, compared to the module currents when discharging into a 1-m $\Omega$  load (Table 1).




Figure 18. Module output at 9 kV with 60- $\mu$ H inductors into a 1-m $\Omega$  load.

Table 1. Summary of relevant modeling results.

Module Charging Voltage (kV)	Module Inductance ( $\mu$ H)	Module Peak Current (kA)	Load Current (MA)	Mass (g)	Load	Exit Velocity (m/s)
9	60	61	1.1	190	Railgun	2260
11	60	73.8	1.33	190	Railgun	2930
9	60	72.3	1.29	NA	1 m $\Omega$	NA
11	60	89	1.60	NA	1 m $\Omega$	NA
9	60	78.7	1.42	9999E03	Short at Breech	NA

## 5. Conclusions

The code successfully modeled the behavior of the in-bore motion and exit velocity of the armature. All the modules were discharged simultaneously (time  $T = 0$ ) for sake of simplicity. The gun current dropped quickly after the maximum value (i.e., from 1.1 MA to 471 kA at exit) (Figure 13). This can be avoided by selecting non-zero firing times for each module.

Future work with the code will be used to model the final pulser's configuration, as nine 116- $\mu$ H inductors will be replaced with nine 30- $\mu$ H inductors, increasing the pulser's output current and shortening its time to reach peak current. This in turn will reflect into higher exit velocities. Other modifications include integrating a remotely operated air-driven module shorting system, presently being built for the nine racks of the 4.5-MJ PPS. This system will insure that energy is discharged from all module capacitors while providing a rugged short to ground, regardless of the eventuality of any fuse opening.

---

## 6. References

---

1. Bhasavanich, D., D. F. Strachan, and R. Ford. "4.5 MJ Modular Pulse Power Supply for ET Gun Applications." Proceedings of the Pulsed Power Conference, Physics International Company, San Leandro, CA, 1993.
2. Hammon, H. G., D. Bhasavanich, and F. T. Warren. "Design Approaches to Pulsed Power Drivers for Electromagnetic and Electrothermal Gun Systems." Proceedings of the Pulsed Power Conference, Physics International Company, San Leandro, CA, 1991.
3. Schroder, K. A. "Study of Possible Power Supply Module Upgrade at the IAT Electromagnetic Research Laboratory." IAT-TM-0051, Institute for Advanced Technology, The University of Texas at Austin, February 2000.

INTENTIONALLY LEFT BLANK.

---

## Appendix A. Netlist of 4.5-MJ Spice Code With Railgun Load

---

### 4.5 MJ PULSER CODE WITH RAILGUN LOAD NETLIST

```
.control
alias VBREECH=V(22)
alias IBREECH=I(V2)
alias VMUZZLE=V(35)
alias VELOCITY=V(31)
alias POSITION=V(32)
alias FORCE=V(33)
alias IPKMOD1=I(VM1)
alias IPKMOD2=I(VM2)
alias IPKMOD3=I(VM3)
alias IPKMOD4=I(VM4)
alias IPKMOD5=I(VM5)
alias IPKMOD6=I(VM6)
alias IPKMOD7=I(VM7)
alias IPKMOD8=I(VM8)
alias IPKMOD9=I(VM9)
alias IPKMOD10=I(VM10)
alias IPKMOD11=I(VM11)
alias IPKMOD12=I(VM12)
alias IPKMOD13=I(VM13)
alias IPKMOD14=I(VM14)
alias IPKMOD15=I(VM15)
alias IPKMOD16=I(VM16)
alias IPKMOD17=I(VM17)
alias IPKMOD18=I(VM18)
save V(22) I(V2) I(VM1)V(35) V(31) V(32) V(33)
save V(22)I(V2)I(VM1)I(VM2)I(VM3)I(VM4)I(VM5)I(VM6)I(VM7)I(VM8)I(VM9)
save V(22)I(V2)I(VM10)I(VM11)I(VM12)I(VM13)I(VM14)I(VM15)I(VM16)I(VM17)I(VM18)
**#SAVE ALL ALLCUR
**#SAVE ALL ALLCUR ALLPOW
.endc
*DEFINE VCHARGE=9K
*DEFINE LPRIME=6.20E-7
*DEFINE MASS=0.190
*DEFINE XSTART=0.3
*DEFINE AREA=2.0E-4
*DEFINE RARM=1.8E-5
*DEFINE VARM=0
*DEFINE INTVEL=0
.OPTIONS TEMP=27 ISCALE=1000000 VSCALE=3000
```

```

.TRAN 10U 10M 0 5U UIC
*.PRINT TRAN I(V2) VBREECH IBREECH VMUZZLE VELOCITY POSITION FORCE
*.PRINT TRAN V(22)V(35)V(31) V(32) V(33)I(V2)I(VM1)I(VM17)
.PRINT TRAN V(31)I(V2)I(VM1)I(VM2)I(VM3)I(VM4)I(VM5)I(VM6)I(VM7)I(VM8)I(VM9)I(VM10)
+I(VM11)I(VM12)I(VM13)I(VM14)I(VM15)I(VM16)I(VM17)I(VM18)
*.PRINT TRAN VBREECH IBREECH VMUZZLE VELOCITY POSITION FORCE I(V2) I(VM1)
* The delay generator determines the banks firing times by closing the switch S1 at each bank
* and putting a voltage of 9kV at nodes 14,24,34,...184 where the V1 voltage source is located
* for each case.
X15 14 24 34 44 54 64 74 84 94 104 114 124 134 144 154 164 174 184 DLYGEN {T1=0.0
+ T2=0.0 T3=0.0U T4=0.0U T5=0.0U T6=0.0U T7=0.0U T8=0.0U T9=0.0U T10=0.0U T11=0.0U
+T12=0.0U T13=0.0U T14=0.0U T15=0.0U T16=0.0U T17=0.0U T18=0.0U}
*****
X11 11 0 14 MODULE1 {V1=VCHARGE}
L1C 11 201 1.2U
VM1 11 201 0
R1C 201 22 3.18M
*****
X12 120 0 24 MODULE2 {V2=VCHARGE}
L2C 120 202 1.2U
VM2 120 202 0
R2C 202 22 3.18M
*****
X13 13 0 34 MODULE3 {V3=VCHARGE}
L3C 13 203 1.2U
VM3 13 203 0
R3C 203 22 3.18M
*****
X14 100 0 44 MODULE4 {V4=VCHARGE}
*** can't use node "14", it inverts the output of this bank, use node "100".
L4C 100 204 1.2U
VM4 100 204 0
R4C 204 22 3.18M
*****
X15 15 0 54 MODULE5 {V5=VCHARGE}
L5C 15 205 1.2U
VM5 15 205 0
R5C 205 22 3.18M
*****
X16 16 0 64 MODULE6 {V6=VCHARGE}
L6C 16 206 1.2U
VM6 16 206 0
R6C 206 22 3.18M
*****
X17 17 0 74 MODULE7 {V7=VCHARGE}
L7C 17 207 1.2U
VM7 17 207 0

```



```

R7C 207 22 3.18M
*****
X18 18 0 84 MODULE8 {V8=VCHARGE}
L8C 18 208 1.2U
VM8 18 208 0
R8C 208 22 3.18M
*****
X1 1 0 94 MODULE9 {V9=VCHARGE}
L9C 1 209 1.2U
VM9 1 209 0
R9C 209 22 3.18M
*****
X2 20 0 104 MODULE10 {V10=VCHARGE}
L10C 20 25 1.2U
VM10 20 25 0
R10C 25 22 3.18M
*****
X3 130 0 114 MODULE11{V11=VCHARGE}
L11C 130 26 1.2U
VM11 130 26 0
R11C 26 22 3.18M
*****
X4 40 0 124 MODULE12 {V12=VCHARGE}
L12C 40 27 1.2U
VM12 40 27 0
R12C 27 22 3.18M
*****
X5 150 0 134 MODULE13 {V13=VCHARGE}
L13C 150 28 1.2U
VM13 150 28 0
R13C 28 22 3.18M
*****
X6 60 0 144 MODULE14 {V14=VCHARGE}
L14C 60 29 1.2U
VM14 60 29 0
R14C 29 22 3.18M
*****
X7 70 0 154 MODULE15 {V15=VCHARGE}
L15C 70 210 1.2U
VM15 70 210 0
R15C 210 22 3.18M
*****
X8 80 0 164 MODULE16 {V16=VCHARGE}
L16C 80 211 1.2U
VM16 80 211 0
R16C 211 22 3.18M
*****

```

X9 90 0 174 MODULE17 {V17=VCHARGE}

L17C 90 212 1.2U

VM17 90 212 0

R17C 212 22 3.18M

\*\*\*\*\*

X10 100 0 184 MODULE18 {V18=VCHARGE}

L18C 100 213 1.2U

VM18 100 213 0

R18C 213 22 3.18M

\*Railgun nodes/ LBC and RBC are the connection's inductance and impedance to the railgun.

LBC 22 239 0.25U

RBC 239 23 .5M

V2 23 30 0

X20 30 0 31 32 33 35 RAILGUN

\*\*\*\*\*

\*\*\*\*\*

.SUBCKT MODULE1 2 4 3 {V1= VCHARGE}

CB 5 4 4242.4U IC={V1}

\*RS is 3M for the Capacitor resistance + 0.65M for the 4 Fuse parallel pack.

RS 5 6 3.65M

LS 6 7 0.3U

D 9 7 DSA908

.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7

M=.5 TT=1E-9)

RCB 10 4 6.65M

LCB 9 10 97N

S1 7 8 3 0 SMOD1 OFF

.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600

LIND 8 12 60U

RIND 12 2 2M

.ENDS

\*\*\*\*\*

.SUBCKT MODULE2 2 4 3 {V2= VCHARGE}

CB 5 4 4239.2U IC={V1}

RS 5 6 3.65M

LS 6 7 0.3U

D 9 7 DSA908

.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7

M=.5 TT=1E-9)

RCB 10 4 6.65M

LCB 9 10 97N

S1 7 8 3 0 SMOD1 OFF

.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600

LIND 8 12 60U

RIND 12 2 2M

.ENDS

\*\*\*\*\*

```

.SUBCKT MODULE3 2 4 3 {V3= VCHARGE}
CB 5 4 4204.2U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****

.SUBCKT MODULE4 2 4 3 {V41= VCHARGE}
CB 5 4 4211.5U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****

.SUBCKT MODULE5 2 4 3 {V5= VCHARGE}
CB 5 4 4216.4U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****

.SUBCKT MODULE6 2 4 3 {V6= VCHARGE}
CB 5 4 4201.7U IC={V1}

```

```

RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS

```

```

*****
.SUBCKT MODULE7 2 4 3 {V7= VCHARGE}
CB 5 4 4239.8U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS

```

```

*****
.SUBCKT MODULE8 2 4 3 {V8= VCHARGE}
CB 5 4 4234.8U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS

```

```

*****
.SUBCKT MODULE9 2 4 3 {V9= VCHARGE}
CB 5 4 4243.8U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U

```

```

D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****

.SUBCKT MODULE10 2 4 3 {V10= VCHARGE}
CB 5 4 4229.6U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****

.SUBCKT MODULE11 2 4 3 {V11= VCHARGE}
CB 5 4 4231.0U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****

.SUBCKT MODULE12 2 4 3 {V12= VCHARGE}
CB 5 4 4205.9U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7

```

```

M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****
.SUBCKT MODULE13 2 4 3 {V13= VCHARGE}
CB 5 4 4216.7U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****
.SUBCKT MODULE14 2 4 3 {V14= VCHARGE}
CB 5 4 4215.4U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M
LCB 9 10 97N
S1 7 8 3 0 SMOD1 OFF
.MODEL SMOD1 SW RON=.3M ROFF=6MEG VT=-2400 VH=2600
LIND 8 12 60U
RIND 12 2 2M
.ENDS
*****
.SUBCKT MODULE15 2 4 3 {V15= VCHARGE}
CB 5 4 4233.7U IC={V1}
RS 5 6 3.65M
LS 6 7 0.3U
D 9 7 DSA908
.MODEL DSA908 D (IS=4.77E-14 N=1.0 BV=13.2E+03 IBV=1.00E-06 RS=0.264E-03 CJO=1.667E-6 VJ=.7
M=.5 TT=1E-9)
RCB 10 4 6.65M

```

## Appendix B. Schematics of 4.5-MJ Spice Code With Railgun Load




Figure B-1. Bank 1/module 1; X<sub>11 11 0 14</sub> calls subckt module 1<sub>243</sub>.




Figure B-2. Bank 1/module 2; X<sub>12 12 0 24</sub> calls subckt module 2<sub>243</sub>.




Figure B-3. Bank 2/module 3;  $X_{13\ 13\ 0\ 34}$  calls subckt module 3<sub>243</sub>.




Figure B-4. Bank 2/module 4;  $X_{14\ 100\ 0\ 44}$  calls subckt module 4<sub>243</sub>.






Figure B-5. Bank 3/module 5; X<sub>15 15 0 54</sub> calls subckt module 5<sub>243</sub>.




Figure B-6. Bank 3/module 6; X<sub>16 16 0 64</sub> calls subckt module 6<sub>243</sub>.




Figure B-7. Bank 4/module 7; X<sub>17</sub> 17 0 74 calls subckt module 7<sub>243</sub>.




Figure B-8. Bank 4/module 8; X<sub>18</sub> 18 0 84 calls subckt module 8<sub>243</sub>.




Figure B-9. Bank 5/module 9;  $X_{11094}$  calls subckt module 9<sub>243</sub>.




Figure B-10. Bank 5/module 10;  $X_{220104}$  calls subckt module 10<sub>243</sub>.




Figure B-11. Bank 6/module 11;  $X_{3\ 130\ 0\ 114}$  calls subckt module 11<sub>243</sub>.




Figure B-12. Bank 6/module 12;  $X_{4\ 40\ 0\ 124}$  calls subckt module 12<sub>243</sub>.




Figure B-13. Bank 7/module 13;  $X_{5\ 150\ 0\ 134}$  calls subckt module 13<sub>243</sub>.




Figure B-14. Bank 7/module 14;  $X_{6\ 60\ 0\ 144}$  calls subckt module 14<sub>243</sub>.




Figure B-15. Bank 8/module 15;  $X_{7\ 70\ 0\ 154}$  calls subckt module 15<sub>243</sub>.




Figure B-16. Bank 8/module 16;  $X_{8\ 80\ 0\ 164}$  calls subckt module 16<sub>243</sub>.




Figure B-17. Bank 9/module 17;  $X_{9\ 90\ 0\ 174}$  calls subckt module 17<sub>243</sub>.




Figure B-18. Bank 9/module 18;  $X_{10\ 100\ 0\ 84}$  calls subckt module 18<sub>243</sub>.




Figure B-19. Railgun outer model B-functions.




Figure B-20. Railgun inner model B-functions.




Figure B-21. Railgun diffusion ladder.



<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
2	DEFENSE TECHNICAL INFORMATION CENTER DTIC OCA 8725 JOHN J KINGMAN RD STE 0944 FT BELVOIR VA 22060-6218
1	HQDA DAMO FDT 400 ARMY PENTAGON WASHINGTON DC 20310-0460
1	OSD OUSD(A&T)/ODDR&E(R) DR R J TREW 3800 DEFENSE PENTAGON WASHINGTON DC 20301-3800
1	COMMANDING GENERAL US ARMY MATERIEL CMD AMCRDA TF 5001 EISENHOWER AVE ALEXANDRIA VA 22333-0001
1	INST FOR ADVNCD TCHNLGY THE UNIV OF TEXAS AT AUSTIN 3925 W BRAKER LN STE 400 AUSTIN TX 78759-5316
1	US MILITARY ACADEMY MATH SCI CTR EXCELLENCE MADN MATH THAYER HALL WEST POINT NY 10996-1786
1	DIRECTOR US ARMY RESEARCH LAB AMSRL D DR D SMITH 2800 POWDER MILL RD ADELPHI MD 20783-1197
1	DIRECTOR US ARMY RESEARCH LAB AMSRL CI AI R 2800 POWDER MILL RD ADELPHI MD 20783-1197

<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
3	DIRECTOR US ARMY RESEARCH LAB AMSRL CI LL 2800 POWDER MILL RD ADELPHI MD 20783-1197
3	DIRECTOR US ARMY RESEARCH LAB AMSRL CI IS T 2800 POWDER MILL RD ADELPHI MD 20783-1197
	<u>ABERDEEN PROVING GROUND</u>
2	DIR USARL AMSRL CI LP (BLDG 305)

<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
1	DIR FOR THE DIRECTORATE OF FORCE DEVELOPMENT US ARMY ARMOR CTR COL E BRYLA FT KNOX KY 40121-5000
1	US ARMY MATERIEL CMD AMC DCG T 5001 EISENHOWER BLVD ALEXANDRIA VA 22333-0001
1	US ARMY MISSILE CMD AMSMI RD DR MCCORKLE REDSTONE ARSENAL AL 35898-5240
2	US ARMY TACOM TARDEC AMSTA TR D MS 207 J CHAPIN M TOURNER WARREN MI 48397-5000
1	US ARMY TACOM ARDEC FSAE GCSS TMA BLDG 354 J BENNETT PICATINNY ARSENAL NJ 07806-5000
6	INST FOR ADVANCED TECH UNIV OF TEXAS AT AUSTIN P SULLIVAN F STEPHANI T WATT W REINCKE D BARNETT M ERENGIL 3925 W BRAKER LN STE 400 AUSTIN TX 78759-531616
2	LOCKHEED MARTIN VOUGHT L FARRIS K COOK PO BOX 650003 MS WT 21 DALLAS TX 75265-0003

<u>NO. OF COPIES</u>	<u>ORGANIZATION</u>
4	UNIV OF TEXAS AT AUSTIN CENTER FOR ELECT A WALLS J KITZMILLER S PRATAP J PAPPAS PRC MAIL CODE R7000 AUSTIN TX 78712
1	INST FOR DEFENSE ANALYSIS I KOHLBERG 1801 N BEAUREGARD ST ALEXANDRIA VA 22311
2	UDLP B GOODELL R JOHNSON MS M170 4800 EAST RIVER RD MINNEAPOLIS MN 55421-1498
1	UNIV OF TEXAS AT AUSTIN M DRIGA ENS 434 DEPT OF ECE MC 60803 AUSTIN TX 78712
1	SAIC K A JAMISON 1247 B N EGLIN PKWY SHALIMAR FL 32579
2	MAXWELL TECHNOLOGIES TITAN PULSED SCIENCES DIV P REIDY T WOLFE 4855 RUFFNER ST SUITE A SAN DIEGO CA 92111
1	MAXWELL PHYSICS INTERNATIONAL C GILMAN 2700 MERCED ST PO BOX 5010 SAN LEANDRO CA 94577-0599

NO. OF  
COPIES ORGANIZATION

ABERDEEN PROVING GROUND

22 DIR USARL  
AMSRL B  
A HORST  
A TANNER  
B FORCH  
D LYON  
AMSRL WM  
E SCHMIDT  
AMSRL WM BC  
P PLOSTINS  
J GARNER  
V OSKAY  
M BUNDY  
G COOPER  
J SAHU  
P WEINACHT  
H EDGE  
B GUIDOS  
A ZIELINSKI  
D WEBB  
K SOENCKSEN  
S WILKERSON  
T ERLINE  
J NEWILL  
M DEL GUERCIO  
AMSRL WM WD  
J POWELL

INTENTIONALLY LEFT BLANK.

REPORT DOCUMENTATION PAGE			Form Approved OMB No. 0704-0188	
<small>Public reporting burden for this collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington, VA 22202-4302, and to the Office of Management and Budget, Paperwork Reduction Project(0704-0188), Washington, DC 20503.</small>				
1. AGENCY USE ONLY (Leave blank)		2. REPORT DATE September 2001		3. REPORT TYPE AND DATES COVERED Final, October 2000-June 2001
4. TITLE AND SUBTITLE A Spice Based Code for ARL's 4.5-MJ Electromagnetic Launcher Pulsed Power Supply System			5. FUNDING NUMBERS 1L1622618.H80	
6. AUTHOR(S) Miguel Del Güercio				
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) U.S. Army Research Laboratory ATTN: AMSRL-WM-BC Aberdeen Proving Ground, MD 21005-5066			8. PERFORMING ORGANIZATION REPORT NUMBER ARL-TR-2592	
9. SPONSORING/MONITORING AGENCY NAMES(S) AND ADDRESS(ES)			10. SPONSORING/MONITORING AGENCY REPORT NUMBER	
11. SUPPLEMENTARY NOTES				
12a. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release; distribution is unlimited.			12b. DISTRIBUTION CODE	
13. ABSTRACT (Maximum 200 words) <p>A Spice-based code (Ispice/4Rx) is used to simulate the U.S. Army Research Laboratory's (ARL's) 4.5-MJ pulsed power supply (PPS) discharge into an EM (electromagnetic) railgun. The code determines the in-bore projectile position, voltage, and current at the breech, and exit current and velocity at the muzzle exit. The code also includes a diffusion model of the railgun rails. The code's primary inputs are the charging voltage to the PPS capacitor banks and the parameters for the different inductive and resistive components of the system which model its hardware. The railgun parameters were chosen to reflect current hardware and program requirements.</p>				
14. SUBJECT TERMS railgun, electromagnetic launcher, electromagnetic railgun			15. NUMBER OF PAGES 50	
			16. PRICE CODE	
17. SECURITY CLASSIFICATION OF REPORT UNCLASSIFIED	18. SECURITY CLASSIFICATION OF THIS PAGE UNCLASSIFIED	19. SECURITY CLASSIFICATION OF ABSTRACT UNCLASSIFIED	20. LIMITATION OF ABSTRACT UL	

INTENTIONALLY LEFT BLANK.